syntax

TYPE type_name IS TABLE OF element_type [NOT NULL]

   INDEX BY key_type; 

Where key_type is one of BINARY_INTEGER, PLS_INTEGER, or VARCHAR2(size_limit). If you choose to index your array by a string, it can be one of VARCHAR2’s subtypes, STRING or LONG. If you use long, it is the same as declaring the key_type to be VARCHAR2(32760). I’ve found it useful and good practice to tie the key type to a column or subtype using the %TYPE attribute.

ordering

Indexing your array by string provides implicit sorting functionality. However, you need to be aware of the lexical sorting order dictated by the character set you are using, especially the nuances of special characters and capital letters.

DECLARE 
   TYPE tan IS TABLE OF PLS_INTEGER INDEX BY VARCHAR2(30);
   lan tan;
   ls_idx VARCHAR2(30);
BEGIN
   lan(' Oompa Loompas') := 199;
   lan('|Oompa Loompas') := 199;
   lan('apple vendors') := 38723;
   lan('Apple seeds') := 10;
   lan('Oompa Loompas') := 199;
   lan('zebra stripes') := 83;
   lan('Zebra herds') := 2342;

   
   msg.p('First',lan.FIRST);
   msg.p('Last',lan.LAST);
   msg.p('');
   
   msg.p('Prior to Apple seeds',lan.PRIOR('Apple seeds'));
   msg.p('Next after Zebra herds',lan.NEXT('Zebra herds'));
   msg.p('');
   
   /*
   FOR s IN lan.FIRST..lan.LAST LOOP --character to number conversion error
      msg.p(lan(s));
   END LOOP;
   */
   

   /* Arrays indexed by string require a different method of iteration */
   ls_idx := lan.FIRST;
   WHILE ls_idx IS NOT NULL LOOP
      msg.p(ls_idx,lan(ls_idx));
      ls_idx := lan.NEXT(ls_idx);
   END LOOP;
END;

/*********************************** OUTPUT ************************************
First:  Oompa Loompas
Last: |Oompa Loompas

Prior to Apple seeds:  Oompa Loompas
Next after Zebra herds: apple vendors

 Oompa Loompas: 199.000
Apple seeds: 10.000
Oompa Loompas: 199.000
Zebra herds: 2,342.000
apple vendors: 38,723.000
zebra stripes: 83.000
|Oompa Loompas: 199.000
*/

performance

“The index-by-varchar2 table is optimized for efficiency of lookup on a non-numeric key, where the notion of sparseness is not really applicable. The index-by-*_integer table (where now *_integer can be either pls_integer or binary_integer) in contrast is optimized for compactness of storage on the assumption that the data is dense. This implies that there might be cases where, even though the key is inherently numeric, it’s better to represent it as a index-by-varchar2 table via a To_Char conversion.” – Bryn Llewellyn

Bryn’s excellent paper “NEW PL/SQL FEATURES IN ACTION: REAL WORLD CASE STUDIES” had tests that showed a twenty-fold speed improvement when using a hash array in favor of SQL lookup against a relational table. My own tests within the context of my current employer’s problem domain revealed a two-fold improvement, which is still significant when you’re dealing with millions of phone call records each day. The improvement was not as great since my tests combined the faster lookup with a shared INSERT task, sucking the life out the speedup as the context switch between the PL/SQL and SQL engines happened for each row.

Here are the results:

lkp_vs_hash_results.xls
Here is the code:

CREATE OR REPLACE PROCEDURE lkp_vs_hash(in_rowlimit IN NUMBER) IS
/* Ideas:
   Make z_trunk_grp an index-organized table. Compare lookup on that to lookup
   on cached table.
   
   Try mimicking hash with array of objects, perform SQL against it.
   
   For bulk array insert, can I get values from lhn in the INSERT's subselect?
   
   For bulk fetch into COR, use cursor that selects from temp_cdr
   and lhn -- instead of having to loop again through the COR to assign the values.
*/
   -- declare type which is a hash of number
   TYPE thn IS TABLE OF NUMBER INDEX BY VARCHAR2(20);
   lhn thn;

   -- local arrays of number
   lan_assoc_id_in t.tan;
   lan_assoc_id_out t.tan;
   lan_cdr_id t.tan;
   
   ln_commit_threshold PLS_INTEGER := 1000;
   
   CURSOR c_cdr(in_limit IN NUMBER) IS 
   SELECT *
   FROM temp_cdr
   WHERE rownum <= in_limit;
   
   TYPE tar_cdr IS TABLE OF temp_cdr%ROWTYPE INDEX BY PLS_INTEGER;
   lar_cdr tar_cdr;
  
   FUNCTION fn_assoc_id (is_trg_cd IN z_trunk_grp.trg_cd%TYPE)
      RETURN z_trunk_grp.assoc_id%TYPE
   IS
      ln_assoc_id z_trunk_grp.assoc_id%TYPE;
   BEGIN
      SELECT assoc_id
      INTO   ln_assoc_id
      FROM   z_trunk_grp
      WHERE  trg_cd = is_trg_cd;
      RETURN ln_assoc_id;
   EXCEPTION
      WHEN NO_DATA_FOUND THEN
         RETURN NULL;
   END fn_assoc_id;
BEGIN
   -- reinitialize target table
   utils.truncate_table('dest_cdr');

   timer.startit('SQL Insert');
      INSERT INTO dest_cdr
      SELECT
       cdr_id
      ,call_dth
      ,switch_id
      ,bgn_dtm
      ,clean_dialed_num
      ,trg_cd_in
      ,fn_assoc_id(trg_cd_in)
      ,trg_cd_out
      ,fn_assoc_id(trg_cd_out)
      ,combo_cd
      FROM temp_cdr;
   timer.stopit('SQL Insert');
   msg.p('Straight update for '||TO_CHAR(in_rowlimit,'FM999G999G999')||' CDR took '||
      timer.fn_elap('SQL Insert'));
      
   
   
   -- reinitialize target table
   utils.truncate_table('dest_cdr');

   timer.startit('Table Lookup');
      FOR lr IN c_cdr(in_rowlimit) LOOP
         lr.assoc_id_in := fn_assoc_id(lr.trg_cd_in);
         lr.assoc_id_out := fn_assoc_id(lr.trg_cd_out);
         
         INSERT INTO dest_cdr VALUES
         (
          lr.cdr_id
         ,lr.call_dth
         ,lr.switch_id
         ,lr.bgn_dtm
         ,lr.clean_dialed_num
         ,lr.trg_cd_in
         ,lr.assoc_id_in
         ,lr.trg_cd_out
         ,lr.assoc_id_out
         ,lr.combo_cd
         );
         
         IF (MOD(c_cdr%ROWCOUNT,ln_commit_threshold) = 0) THEN
            COMMIT;
         END IF;
      END LOOP;
   timer.stopit('Table Lookup');
   msg.p('Table Lookup for '||TO_CHAR(in_rowlimit,'FM999G999G999')||' CDR took '
      ||timer.fn_elap('Table Lookup'));  

   
   
   -- reinitialize target table
   utils.truncate_table('dest_cdr');

   timer.startit('Array Lookup');

      timer.startit('Load Array');
         FOR lr IN (SELECT trg_cd, assoc_id FROM z_trunk_grp) LOOP
            -- can't bulk collect into index-by-string arrays yet, so we loop
            lhn(lr.trg_cd) := lr.assoc_id;
         END LOOP;
      timer.stopit('Load Array');
      msg.p('Array population with codes took '||timer.fn_elap('Load Array'));
   
      FOR lr IN c_cdr(in_rowlimit) LOOP
         lr.assoc_id_in := lhn(lr.trg_cd_in);
         lr.assoc_id_out := lhn(lr.trg_cd_out);
         
         INSERT INTO dest_cdr VALUES
         (
          lr.cdr_id
         ,lr.call_dth
         ,lr.switch_id
         ,lr.bgn_dtm
         ,lr.clean_dialed_num
         ,lr.trg_cd_in
         ,lr.assoc_id_in
         ,lr.trg_cd_out
         ,lr.assoc_id_out
         ,lr.combo_cd
         );
         
         IF (MOD(c_cdr%ROWCOUNT,ln_commit_threshold) = 0) THEN
            COMMIT;
         END IF;
      END LOOP;
   timer.stopit('Array Lookup');
   msg.p('Array Lookup for '||TO_CHAR(in_rowlimit,'FM999G999G999')||' CDR took '||
      timer.fn_elap('Array Lookup'));  

   -- reinitialize target table
   utils.truncate_table('dest_cdr');

   timer.startit('Array Lookup w/Record Insert');   
      FOR lr IN c_cdr(in_rowlimit) LOOP
         lr.assoc_id_in := lhn(lr.trg_cd_in);
         lr.assoc_id_out := lhn(lr.trg_cd_out);
         
         INSERT INTO dest_cdr VALUES lr;
         
         IF (MOD(c_cdr%ROWCOUNT,ln_commit_threshold) = 0) THEN
            COMMIT;
         END IF;
      END LOOP;
   timer.stopit('Array Lookup w/Record Insert');
   msg.p('Array Lookup for '||TO_CHAR(in_rowlimit,'FM999G999G999')||' CDR took '||
      timer.fn_elap('Array Lookup w/Record Insert'));  
   
   -- reinitialize target table
   utils.truncate_table('dest_cdr');

   timer.startit('Array Lookup w/Bulk Array Insert');
      FOR lr IN c_cdr(in_rowlimit) LOOP
         -- speed if I use a counter instead of function for index?
         lan_cdr_id(c_cdr%ROWCOUNT) := lr.cdr_id;
         lan_assoc_id_in(c_cdr%ROWCOUNT) := lhn(lr.trg_cd_in);
         lan_assoc_id_out(c_cdr%ROWCOUNT) := lhn(lr.trg_cd_out);      
      END LOOP;
      
      FORALL i IN lan_cdr_id.FIRST..lan_cdr_id.LAST
         INSERT INTO dest_cdr
         SELECT
          cdr_id
         ,call_dth
         ,switch_id
         ,bgn_dtm
         ,clean_dialed_num
         ,trg_cd_in
         ,lan_assoc_id_in(i)
         ,trg_cd_out
         ,lan_assoc_id_out(i)
         ,combo_cd
         FROM temp_cdr
         WHERE cdr_id = lan_cdr_id(i);
      
   timer.stopit('Array Lookup w/Bulk Array Insert');
   msg.p('Array Lookup for '||TO_CHAR(in_rowlimit,'FM999G999G999')||' CDR took '||
      timer.fn_elap('Array Lookup w/Bulk Array Insert'));  

   -- reinitialize target table
   utils.truncate_table('dest_cdr');
   
   timer.startit('Array Lookup w/Bulk Record Insert');
      OPEN c_cdr(in_rowlimit);
      FETCH c_cdr BULK COLLECT INTO lar_cdr LIMIT in_rowlimit;
      CLOSE c_cdr;
      
      /*FOR i IN lar_cdr.FIRST..lar_cdr.LAST LOOP
         -- to surmout no in-bind references limitation in bulk dml with records
         lan_cdr_id(i) := lar_cdr(i).cdr_id;
         
         lar_cdr(i).assoc_id_in := lhn(lar_cdr(i).trg_cd_in);
         lar_cdr(i).assoc_id_out := lhn(lar_cdr(i).trg_cd_out);
      END LOOP;
      
      FORALL i IN lan_cdr_id.FIRST..lan_cdr_id.LAST
         UPDATE temp_cdr
         SET    ROW = lar_cdr(i)
         WHERE  cdr_id = lan_cdr_id(i);
      */
      FOR i IN lar_cdr.FIRST..lar_cdr.LAST LOOP
         lar_cdr(i).assoc_id_in := lhn(lar_cdr(i).trg_cd_in);
         lar_cdr(i).assoc_id_out := lhn(lar_cdr(i).trg_cd_out);
      END LOOP;
         
      FORALL i IN lar_cdr.FIRST..lar_cdr.LAST
         INSERT INTO dest_cdr VALUES lar_cdr(i);
      
   timer.stopit('Array Lookup w/Bulk Record Insert');
   msg.p('Array Lookup for '||TO_CHAR(in_rowlimit,'FM999G999G999')||' CDR took '||
      timer.fn_elap('Array Lookup w/Bulk Record Insert'));  

END lkp_vs_hash;

